



# Shaders


## Part 1



# The GPU pipeline




# The GPU pipeline




# The vertex shader


A program that transforms the attributes of  
every vertex passed to the GPU  
and passes data to the fragment shader.

The vertex shader is where the model, view and projection matrices are applied to the vertex.











# The fragment shader

A program that returns the color of each pixel  
when geometry is rasterized on the GPU.







**2K**  
CZECH  
ARTTEST  
2048x2048 TEXTURES  
2754 TRIS  
[WWW.ALEXANDER.DELAGRANGE.COM](http://WWW.ALEXANDER.DELAGRANGE.COM)



2048x2048 TEXTURES  
2754 TRIS  
[WWW.ALEXANDER.DELAGRANGE.COM](http://WWW.ALEXANDER.DELAGRANGE.COM)








4 4 4

\$ 0





XBOX

The final shader program is a combination  
of a vertex and a fragment shader.

# GLSL

## OpenGL Shading Language

# Anatomy of a simple GLSL vertex shader.

```
attribute vec4 position;
attribute vec2 texCoord;

uniform mat4 modelMatrix;
uniform mat4 viewMatrix;
uniform mat4 projectionMatrix;

varying vec2 texCoordVar;

void main()
{
    vec4 p = viewMatrix * modelMatrix * position;
    texCoordVar = texCoord;
    gl_Position = projectionMatrix * p;
}
```

# Anatomy of a simple GLSL vertex shader.

```
attribute vec4 position;  
attribute vec2 texCoord;
```

## ATTRIBUTES

```
uniform mat4 modelMatrix;  
uniform mat4 viewMatrix;  
uniform mat4 projectionMatrix;
```

## UNIFORMS

```
varying vec2 texCoordVar;
```

## VARYING VARIABLES

```
void main()  
{  
    vec4 p = viewMatrix * modelMatrix * position;  
    texCoordVar = texCoord;  
    gl_Position = projectionMatrix * p;  
}
```

# Anatomy of a simple GLSL fragment shader.

```
uniform sampler2D diffuse;  
  
varying vec2 texCoordVar;  
  
void main() {  
    gl_FragColor = texture2D(diffuse, texCoordVar);  
}
```

# Anatomy of a simple GLSL fragment shader.

```
uniform sampler2D diffuse;  
  
varying vec2 texCoordVar;  
  
void main() {  
    gl_FragColor = texture2D(diffuse, texCoordVar);  
}
```

**UNIFORMS**

**VARYING VARIABLES**

# Using shaders in OpenGL.

Loading and compiling vertex  
and fragment shaders.

# vertex\_shader.glsl

```
attribute vec4 position;
attribute vec2 texCoord;

uniform mat4 modelMatrix;
uniform mat4 viewMatrix;
uniform mat4 projectionMatrix;

varying vec2 texCoordVar;

void main()
{
    vec4 p = viewMatrix * modelMatrix * position;
    texCoordVar = texCoord;
    gl_Position = projectionMatrix * p;
}
```

# fragment\_shader.glsl

```
uniform sampler2D diffuse;  
  
varying vec2 texCoordVar;  
  
void main() {  
    gl_FragColor = texture2D(diffuse, texCoordVar);  
}
```

# Load our shader files into an `std::string`.

```
std::ifstream infile("vertex_shader.frag");
if(infile.fail()) {
    std::cout << "Error opening shader file" << std::endl;
}
std::stringstream buffer;
buffer << infile.rdbuf();
std::string vertexShader = buffer.str();
```

# Creating and compiling a shader from string.

# Use GL\_VERTEX\_SHADER for vertex shader.

```
GLuint vertexShaderID = glCreateShader(GL_VERTEX_SHADER);

const char *vertexShaderString = vertexShader.c_str();
int vertexShaderStringLength = vertexShader.size();

glShaderSource(vertexShaderID, 1, &vertexShaderString, & vertexShaderStringLength);
glCompileShader(vertexShaderID);
```

# Use GL\_FRAGMENT\_SHADER for fragment shader.

```
GLuint fragmentShaderID = glCreateShader(GL_FRAGMENT_SHADER);

const char *fragmentShaderString = fragmentShader.c_str();
int fragmentShaderStringLength = fragmentShader.size();

glShaderSource(fragmentShaderID, 1, &fragmentShaderString, &fragmentShaderStringLength);
glCompileShader(fragmentShaderID);
```

# Creating and linking the shader program from the fragment and vertex shaders.

```
GLuint exampleProgram = glCreateProgram();
glAttachShader(exampleProgram, vertexShaderID);
glAttachShader(exampleProgram, fragmentShaderID);
glLinkProgram(exampleProgram);
```

Get references to the **attribute** and **uniform** locations of the shader program so we can pass data to it.

```
attribute vec4 position;  
attribute vec2 texCoord;
```

**ATTRIBUTES**

```
uniform mat4 modelMatrix;  
uniform mat4 viewMatrix;  
uniform mat4 projectionMatrix;
```

**UNIFORMS**

# Uniforms

```
GLint projectionMatrixUniform = glGetUniformLocation(exampleProgram, "projectionMatrix");
GLint modelMatrixUniform = glGetUniformLocation(exampleProgram, "modelMatrix");
GLint viewMatrixUniform = glGetUniformLocation(exampleProgram, "viewMatrix");
```

# Attributes

```
GLuint positionAttribute = glGetAttribLocation(exampleProgram, "position");
GLuint texCoordAttribute = glGetAttribLocation(exampleProgram, "texCoord");
```

Drawing using our  
shader program.

# Tell OpenGL to use our program.

```
glUseProgram(exampleProgram);
```

# Bind our model, view and projection matrices.

```
glUniformMatrix4fv(projectionMatrixUniform, 1, GL_FALSE, projectionMatrix.ml);
glUniformMatrix4fv(modelMatrixUniform, 1, GL_FALSE, modelMatrix.ml);
glUniformMatrix4fv(viewMatrixUniform, 1, GL_FALSE, viewMatrix.ml);
```

# Bind shader attributes.

```
float vertices[] = {-0.5, 0.5, -0.5, -0.5, 0.5, -0.5, -0.5, 0.5, 0.5, -0.5, 0.5, 0.5};  
glVertexAttribPointer(positionAttribute, 2, GL_FLOAT, false, 0, vertices);  
 glEnableVertexAttribArray(positionAttribute);  
  
float texCoords[] = {0.0, 0.0, 0.0, 1.0, 1.0, 1.0, 0.0, 0.0, 1.0, 1.0, 1.0, 0.0};  
glVertexAttribPointer(texCoordAttribute, 2, GL_FLOAT, false, 0, texCoords);  
 glEnableVertexAttribArray(texCoordAttribute);  
  
glDrawArrays(GL_TRIANGLES, 0, 6);
```

Catching GLSL errors when  
compiling shaders.

After calling `glCompileShader` you can check if it failed to compile and get a readable error message.

```
GLint compileSuccess;
glGetShaderiv(vertexShaderID, GL_COMPILE_STATUS, &compileSuccess);
if (compileSuccess == GL_FALSE) {

    GLchar messages[256];
    glGetShaderInfoLog(shaderID, sizeof(messages), 0, &messages[0]);
    std::cout << messages << std::endl;
}
```

# Intro to basic GLSL

## Basic Types

<b>void</b>	no function return value or empty parameter list
<b>bool</b>	Boolean
<b>int</b>	signed integer
<b>float</b>	floating scalar
<b>vec2, vec3, vec4</b>	n-component floating point vector
<b>bvec2, bvec3, bvec4</b>	Boolean vector
<b>ivec2, ivec3, ivec4</b>	signed integer vector
<b>mat2, mat3, mat4</b>	2x2, 3x3, 4x4 float matrix
<b>sampler2D</b>	access a 2D texture
<b>samplerCube</b>	access cube mapped texture

Vertex shader must set **gl\_Position** (which is a homogeneous **vec4** coordinate) to set the final position of the vertex being drawn.

If it is passing any varying data to the fragment shader, it must set those as well.

```
attribute vec4 position;
attribute vec2 texCoord;

uniform mat4 modelMatrix;
uniform mat4 viewMatrix;
uniform mat4 projectionMatrix;

varying vec2 texCoordVar;

void main()
{
    vec4 p = viewMatrix * modelMatrix * position;
    texCoordVar = texCoord;
    gl_Position = projectionMatrix * p;
}
```

Fragment shader must set **gl\_FragColor** (which is a **vec4** RGBA color value) to set the final color of the pixel being rendered.

The **texture2D** function is built in and takes a **sampler2D** texture and a **vec2** UV coordinate and returns a **vec4** RGBA color from that texture at that coordinate.

```
uniform sampler2D diffuse;  
  
varying vec2 texCoordVar;  
  
void main() {  
    gl_FragColor = texture2D(diffuse, texCoordVar);  
}
```

# Simple shader examples.

# Inverting texture color.

```
uniform sampler2D texture;  
  
varying vec2 texCoordVar;  
  
void main()  
{  
    vec4 finalColor = 1.0 - texture2D( texture, texCoordVar);  
    finalColor.a = texture2D( texture, texCoordVar).a;  
    gl_FragColor = finalColor;  
}
```

# Making a texture black and white.

```
uniform sampler2D texture;  
  
varying vec2 texCoordVar;  
  
void main()  
{  
    vec4 texColor = texture2D( texture, texCoordVar);  
    vec4 finalColor = vec4((texColor.r + texColor.g + texColor.b)/3.0);  
    finalColor.a = texture2D( texture, texCoordVar).a;  
    gl_FragColor = finalColor;  
}
```

# Functions in GLSL

## Built-In Functions

### Angle & Trigonometry Functions [8.1]

Component-wise operation. Parameters specified as *angle* are assumed to be in units of radians. T is float, vec2, vec3, vec4.

T <b>radians</b> (T <i>degrees</i> )	degrees to radians
T <b>degrees</b> (T <i>radians</i> )	radians to degrees
T <b>sin</b> (T <i>angle</i> )	sine
T <b>cos</b> (T <i>angle</i> )	cosine
T <b>tan</b> (T <i>angle</i> )	tangent
T <b>asin</b> (T <i>x</i> )	arc sine
T <b>acos</b> (T <i>x</i> )	arc cosine
T <b>atan</b> (T <i>y</i> , T <i>x</i> )	arc tangent
T <b>atan</b> (T <i>y_over_x</i> )	

### Exponential Functions [8.2]

Component-wise operation. T is float, vec2, vec3, vec4.

T <b>pow</b> (T <i>x</i> , T <i>y</i> )	$x^y$
T <b>exp</b> (T <i>x</i> )	$e^x$
T <b>log</b> (T <i>x</i> )	$\ln$
T <b>exp2</b> (T <i>x</i> )	$2^x$
T <b>log2</b> (T <i>x</i> )	$\log_2$
T <b>sqrt</b> (T <i>x</i> )	square root
T <b>inversesqrt</b> (T <i>x</i> )	inverse square root

### Common Functions [8.3]

Component-wise operation. T is float, vec2, vec3, vec4.

T <b>abs</b> (T <i>x</i> )	absolute value
T <b>sign</b> (T <i>x</i> )	returns -1.0, 0.0, or 1.0
T <b>floor</b> (T <i>x</i> )	nearest integer $\leq x$
T <b>ceil</b> (T <i>x</i> )	nearest integer $\geq x$
T <b>fract</b> (T <i>x</i> )	$x - \text{floor}(x)$
T <b>mod</b> (T <i>x</i> , T <i>y</i> )	modulus
T <b>mod</b> (T <i>x</i> , float <i>y</i> )	
T <b>min</b> (T <i>x</i> , T <i>y</i> )	minimum value
T <b>min</b> (T <i>x</i> , float <i>y</i> )	
T <b>max</b> (T <i>x</i> , T <i>y</i> )	maximum value
T <b>max</b> (T <i>x</i> , float <i>y</i> )	
T <b>clamp</b> (T <i>x</i> , T <i>minVal</i> , T <i>maxVal</i> )	
T <b>clamp</b> (T <i>x</i> , float <i>minVal</i> , float <i>maxVal</i> )	$\min(\max(x, \text{minVal}), \text{maxVal})$
T <b>mix</b> (T <i>x</i> , T <i>y</i> , T <i>a</i> )	linear blend of <i>x</i> and <i>y</i>
T <b>mix</b> (T <i>x</i> , T <i>y</i> , float <i>a</i> )	
T <b>step</b> (T <i>edge</i> , T <i>x</i> )	0.0 if $x < \text{edge}$ , else 1.0
T <b>step</b> (float <i>edge</i> , T <i>x</i> )	
T <b>smoothstep</b> (T <i>edge0</i> , T <i>edge1</i> , T <i>x</i> )	
T <b>smoothstep</b> (float <i>edge0</i> , float <i>edge1</i> , T <i>x</i> )	clip and smooth

### Geometric Functions [8.4]

These functions operate on vectors as vectors, not component-wise. T is float, vec2, vec3, vec4.

float <b>length</b> (T <i>x</i> )	length of vector
float <b>distance</b> (T <i>p0</i> , T <i>p1</i> )	distance between points
float <b>dot</b> (T <i>x</i> , T <i>y</i> )	dot product
vec3 <b>cross</b> (vec3 <i>x</i> , vec3 <i>y</i> )	cross product
T <b>normalize</b> (T <i>x</i> )	normalize vector to length 1
T <b>faceforward</b> (T <i>N</i> , T <i>I</i> , T <i>Nref</i> )	returns <i>N</i> if $\text{dot}(N_{\text{ref}}, I) < 0$ , else - <i>N</i>
T <b>reflect</b> (T <i>I</i> , T <i>N</i> )	reflection direction $I - 2 * \text{dot}(N, I) * N$
T <b>refract</b> (T <i>I</i> , T <i>N</i> , float <i>eta</i> )	refraction vector

### Matrix Functions [8.5]

Type mat is any matrix type.

mat <b>matrixCompMult</b> (mat <i>x</i> , mat <i>y</i> )	multiply <i>x</i> by <i>y</i> component-wise
--	--

### Vector Relational Functions [8.6]

Compare *x* and *y* component-wise. Sizes of input and return vectors for a particular call must match. Type bvec is bvecn; vec is vecn; ivec is ivec $n$  (where  $n$  is 2, 3, or 4). T is the union of vec and ivec.

bvec <b>lessThan</b> (T <i>x</i> , T <i>y</i> )	$x < y$
bvec <b>lessThanEqual</b> (T <i>x</i> , T <i>y</i> )	$x \leq y$
bvec <b>greaterThan</b> (T <i>x</i> , T <i>y</i> )	$x > y$
bvec <b>greaterThanEqual</b> (T <i>x</i> , T <i>y</i> )	$x \geq y$
bvec <b>equal</b> (T <i>x</i> , T <i>y</i> )	$x == y$
bvec <b>equal</b> (bvec <i>x</i> , bvec <i>y</i> )	
bvec <b>notEqual</b> (T <i>x</i> , T <i>y</i> )	$x != y$
bvec <b>notEqual</b> (bvec <i>x</i> , bvec <i>y</i> )	
bool <b>any</b> (bvec <i>x</i> )	true if any component of <i>x</i> is true
bool <b>all</b> (bvec <i>x</i> )	true if all components of <i>x</i> are true
bvec <b>not</b> (bvec <i>x</i> )	logical complement of <i>x</i>

### Texture Lookup Functions [8.7]

Available only in vertex shaders.

vec4 <b>texture2DLod</b> (sampler2D <i> sampler</i> , vec2 <i>coord</i> , float <i>lod</i> )	
vec4 <b>texture2DProjLod</b> (sampler2D <i> sampler</i> , vec3 <i>coord</i> , float <i>lod</i> )	
vec4 <b>texture2DProjLod</b> (sampler2D <i> sampler</i> , vec4 <i>coord</i> , float <i>lod</i> )	
vec4 <b>textureCubeLod</b> (samplerCube <i> sampler</i> , vec3 <i>coord</i> , float <i>lod</i> )	

Available only in fragment shaders.

vec4 <b>texture2D</b> (sampler2D <i> sampler</i> , vec2 <i>coord</i> , float <i>bias</i> )	
vec4 <b>texture2DProj</b> (sampler2D <i> sampler</i> , vec3 <i>coord</i> , float <i>bias</i> )	
vec4 <b>texture2DProj</b> (sampler2D <i> sampler</i> , vec4 <i>coord</i> , float <i>bias</i> )	
vec4 <b>textureCube</b> (samplerCube <i> sampler</i> , vec3 <i>coord</i> , float <i>bias</i> )	

Available in vertex and fragment shaders.

vec4 <b>texture2D</b> (sampler2D <i> sampler</i> , vec2 <i>coord</i> )	
vec4 <b>texture2DProj</b> (sampler2D <i> sampler</i> , vec3 <i>coord</i> )	
vec4 <b>texture2DProj</b> (sampler2D <i> sampler</i> , vec4 <i>coord</i> )	
vec4 <b>textureCube</b> (samplerCube <i> sampler</i> , vec3 <i>coord</i> )	

# Saturation function example.

```
uniform sampler2D texture;
varying vec2 texCoordVar;

vec3 saturation_func(vec3 rgb, float adjustment)
{
    const vec3 W = vec3(0.2125, 0.7154, 0.0721);
    vec3 intensity = vec3(dot(rgb, W));
    return mix(intensity, rgb, adjustment);
}

void main()
{
    vec4 finalColor;
    finalColor.rgb = saturation_func(texture2D( texture, texCoordVar).rgb, 2.0);
    finalColor.a = texture2D( texture, texCoordVar).a;
    gl_FragColor = finalColor;
}
```

# Passing variables as uniforms.

# Example: passing the saturation value from our C++ code.

```
uniform sampler2D texture;
uniform float saturationAmount;
varying vec2 texCoordVar;

vec3 saturation_func(vec3 rgb, float adjustment)
{
    const vec3 W = vec3(0.2125, 0.7154, 0.0721);
    vec3 intensity = vec3(dot(rgb, W));
    return mix(intensity, rgb, adjustment);
}

void main()
{
    vec4 finalColor;
    finalColor.xyz = saturation_func(texture2D( texture, texCoordVar).xyz, saturationAmount);
    finalColor.a = texture2D( texture, texCoordVar).a;
    gl_FragColor = finalColor;
}
```

# Get the saturation amount uniform location.

```
GLint saturationAmountUniform = glGetUniformLocation(exampleProgram, "saturationAmount");
```

# Before rendering, bind a value to it.

```
glUniform1f(saturationAmountUniform, 2.0f);
```

# Some GLSL types and their corresponding C++ uniform binding functions.

**float** - **glUniform1f**(location, value);

**vec2** - **glUniform2f**(location, value1, value2);

**vec3** - **glUniform3f**(location, value1, value2, value3);

**vec4** - **glUniform4f**(location, value1, value2, value3, value4);

# Example: scrolling a texture.

```
uniform sampler2D texture;
uniform vec2 scroll;
varying vec2 texCoordVar;

void main()
{
    gl_FragColor = texture2D( texture, texCoordVar + scroll);
}
```

# Get the scroll amount uniform location.

```
GLint scrollUniform = glGetUniformLocation(exampleProgram, "scroll");
```

Before rendering, bind a value to it.

```
glUniform2f(scrollUniform, ticks, 0.0f);
```